A numerical study of the stationary solution of the Keller-Segel model in self-similar variables

```
ϵ = 10^-6;
R = 5;
```

The density

```
F[a_, DS_] := Plot[Exp[-1/2 r^2 + v[r]] /.
   NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0}, {v, v'}, {s, ϵ, R}],
  {r, ϵ, R}, DisplayFunction → DS, PlotRange → {Automatic, All}]
```

```
{F[0.1, $DisplayFunction], F[1, $DisplayFunction], F[5, $DisplayFunction]}
```




```
Show[Table[F[a, Identity], {a, -2.3, 2.5, 0.3}],
 DisplayFunction → $DisplayFunction, PlotRange → {{0, 2}, All}]
```



The mass distribution

```
F[a_, DS_] := Plot[m[r] /.
   NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]], m'[s] == s Exp[-1/2 s^2 + v[s]], v[ϵ] == a,
     v'[ϵ] == 0, m[ϵ] == 0}, {v, v', m}, {s, ϵ, R}], {r, ϵ, R}, DisplayFunction → DS]
```

```
{F[0.1, $DisplayFunction], F[1, $DisplayFunction], F[10, $DisplayFunction]}
```



The normalization constant

```
F[a_, DS_] := Plot[v[r] + m[R] Log[r] /.
    NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]], m'[s] == s Exp[-1/2 s^2 + v[s]], v[ϵ] == a,
      v'[ϵ] == 0, m[ϵ] == 0}, {v, v', m}, {s, ϵ, R}], {r, ϵ, R}, DisplayFunction → DS]
{F[0.1, $DisplayFunction], F[1, $DisplayFunction], F[10, $DisplayFunction]}
```



The bifurcation diagram

```
ParametricPlot[{m[R], a - v[R] - m[R] Log[R]} /.
   NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]], m'[s] == s Exp[-1/2 s^2 + v[s]], v[ϵ] == a,
     v'[ϵ] == 0, m[ϵ] == 0}, {v, v', m}, {s, ϵ, R}], {a, 0.1, 10}, AspectRatio → 0.7]
```



The range of the mass and its dependence in a

```
Mass[a_] :=
  2 π m[R] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]], m'[s] == s Exp[-1/2 s^2 + v[s]],
      v[ϵ] == a, v'[ϵ] == 0, m[ϵ] == 0}, {v, v', m}, {s, ϵ, R}][[1]]
```

```
Plot[Mass[a]/(8 π), {a, -5, 10}]
```



Plots of of c_a in terms of a

```
Show[Table[Plot[v[r] - v[R] - m[R] Log[R] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
    m'[s] == s Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0, m[ϵ] == 0}, {v, v', m}, {s, ϵ, R}],
   {r, ϵ, R}, DisplayFunction → Identity], {a, -2.3, 2.5, 0.3}],
 DisplayFunction → $DisplayFunction, PlotRange → {{0, 3}, {-1, 1.6}}]
```



The asymptotic regime as a→+∞

```
b[a_] := v[R] + m[R] Log[R] /.
  NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]], m'[s] == s Exp[-1/2 s^2 + v[s]],
     v[ϵ] == a, v'[ϵ] == 0, m[ϵ] == 0}, {v, v', m}, {s, ϵ, R}][[1]]
F[a_, DS_] := Module[{λ = Sqrt[8] ⅇ^(-a/2)}, Plot[λ^2 Exp[-1/2 (λ r)^2 + v[λ r]] /. NDSolve[
      {-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0}, {v, v'}, {s, ϵ, R}],
    {r, ϵ, R/λ}, DisplayFunction → DS, PlotRange → {Automatic, All}]]
```

```
Show[Table[F[a, Identity], {a, -2, 5, 0.5}],
 DisplayFunction → $DisplayFunction, PlotRange → {{0, 2}, All}]
```

$$\text{Show}\left[\%, \text{Plot}\left[\frac{8}{(1+r^2)^2}, \{r, 0, 2.2\}, \text{PlotStyle} \to \text{Red}\right], \text{PlotRange} \to \{\{0, 2\}, \text{All}\}\right]$$





The asymptotic regime as a→-∞

$$\text{Integrate}\left[\frac{1 - e^{-s}}{s}, \{s, 0, x\}, \text{Assumptions} \to \text{Re}[x] > 0\right]$$

$$\psi[r\_] := \frac{1}{2}\,(\text{EulerGamma} + \text{Gamma}[0, x] + \text{Log}[x]) \;/.\; x \to \frac{1}{2} r^2$$

```
EulerGamma + Gamma[0, x] + Log[x]
```

$$F[a\_, DS\_] := \text{Plot}\Big[e^{-a}\,(a - v[r]) \;/.\; \text{NDSolve}\left[\left\{-v''[s] - \frac{v'[s]}{s} == \text{Exp}\left[-\frac{1}{2} s^2 + v[s]\right], v[\epsilon] == a, v'[\epsilon] == 0\right\}, \{v, v'\}, \{s, \epsilon, R\}\right], \{r, \epsilon, R\}, \text{DisplayFunction} \to DS, \text{PlotRange} \to \{\text{Automatic}, \text{All}\}\Big]$$

```
Show[Table[F[a, Identity], {a, -5, 5, 0.5}], DisplayFunction → $DisplayFunction]
Show[%, Plot[ψ[r], {r, 0, R}, PlotStyle → Red]]
```









The kernel

```
F[a_, DS_] := Plot[f[r] /.
    NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]], -f''[s] - f'[s]/s == Exp[-1/2 s^2 + v[s]] f[s],
      v[ϵ] == a, v'[ϵ] == 0, f[ϵ] == 1, f'[ϵ] == 0}, {v, v', f, f'}, {s, ϵ, R}],
   {r, ϵ, R}, DisplayFunction → DS, PlotRange → {Automatic, All}]

{F[0.1, $DisplayFunction], F[1, $DisplayFunction], F[5, $DisplayFunction]}
```




```
R = 10;
Show[Table[F[a, Identity], {a, -3, 10, 0.5}],
 DisplayFunction → $DisplayFunction, AxesOrigin → {0, 0}]
```




```
R = 5;

F[a_, DS_] := Plot[f[r] Exp[-1/2 r^2 + v[r]] /.
    NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]], -f''[s] - f'[s]/s == Exp[-1/2 s^2 + v[s]] f[s],
      v[ϵ] == a, v'[ϵ] == 0, f[ϵ] == 1, f'[ϵ] == 0}, {v, v', f, f'}, {s, ϵ, R}],
   {r, ϵ, R}, DisplayFunction → DS, PlotRange → {Automatic, All}]

{F[0.1, $DisplayFunction], F[1, $DisplayFunction], F[5, $DisplayFunction]}
```



```
Show[Table[F[a, Identity], {a, -3, 10, 0.5}],
 DisplayFunction → $DisplayFunction, PlotRange → {{0, 1.5}, {-4, 14}}]
```



The radial eigenvalues

```
ϵ = 10^-6;
R = 7;
F[a_, λmin_, λmax_] :=
 Plot[Log[1 + f[R]^2] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
      -f''[s] - f'[s]/s == (v'[s] - s) (f'[s] - φ[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
      φ'[s] == -φ[s]/s - f[s] Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0,
      f'[ϵ] == 0, f[ϵ] == 1, φ[ϵ] == 0}, {v, v', f, f', φ}, {s, ϵ, R}],
  {λ, λmin, λmax}, PlotRange → {Automatic, All}]
```

```
F[1, -1, 10]
```






```
{F[0, -1, 10], F[1, -1, 10], F[5, -1, 10]}
```



The eigenfunction and the shooting method around λ=2

```
R = 5;
FP[a_, λ_] := Plot[{Exp[-1/2 r^2 + v[r]], f[r] Exp[-1/2 r^2 + v[r]]} /.
   NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
     -f''[s] - f'[s]/s == (v'[s] - s) (f'[s] - φ[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
     φ'[s] == -φ[s]/s - f[s] Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0, f'[ϵ] == 0,
     f[ϵ] == 1, φ[ϵ] == 0}, {v, v', f, f', φ}, {s, ϵ, R}], {r, ϵ, R}, PlotRange → All]
```

```
{FP[1, 2], FP[1, 1.9], FP[1, 2.1]}
```



```
{FP[0.1, 2], FP[0.1, 1.9], FP[0.1, 2.1]}
```



```
FP[1, 2]
```



Computation of the spectrum of the operator restricted to radial functions

```
η = 10^-8;
R = 7;
F[a_, λ_] := Log[1 + f[R]^2] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
     -f''[s] - f'[s]/s == (v'[s] - s) (f'[s] - φ[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
     φ'[s] == -φ[s]/s - f[s] Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0,
     f'[ϵ] == 0, f[ϵ] == 1, φ[ϵ] == 0}, {v, v', f, f', φ}, {s, ϵ, R}][[1]]
Fiter[a_, λ_, h_, b_] := If[Abs[h] < η, {λ, h}, Module[{m = F[a, λ]},
   If[(m - b) < 0, Fiter[a, λ + h, h, m], Fiter[a, λ - h / 2, -h / 2, m]]]]
Search[a_, λ_, h_] := Fiter[a, λ, h, F[a, λ - h]]
Bifurcation[λ_, h_] := ParametricPlot[{Mass[a], Search[a, λ, h][[1]]}, {a, -3, 10}]
```

```
P1 = Bifurcation[1, 0.2];
Show[P1, AspectRatio → 1]
```




```
P2 = Bifurcation[3, 0.2];
Show[P2, AspectRatio → 1]
```



```
P3 = Bifurcation[5.5, 0.2];
Show[P3, AspectRatio → 1]
```




```
Show[{P1, P2, P3}, PlotRange → {{0, 8 π}, {0, 7.5}}, AspectRatio → 1, AxesOrigin → {0, 0}]
```



Radial eigenfunctions

```
Mass[1]
Search[1, 3, 0.2][[1]]
Show[FP[1, %], PlotRange → {All, {-0.45, 0.1}}]
Search[1, 5.5, 0.2][[1]]
Show[FP[1, %], PlotRange → {All, {-0.1, 0.1}}]
```

9.10875

4.1944



6.27881



Second eigenfunction

```
ϵ = 10^-7;
R = 5;
F1[a_, λ_] := Plot[r/(f[r] - 1) (v'[r] - r) /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
     -f''[s] - f'[s]/s == (v'[s] - s) (f'[s] - φ[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
     φ'[s] == -φ[s]/s - f[s] Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0, f'[ϵ] == 0, f[ϵ] == 1,
     φ[ϵ] == 0}, {v, v', f, f', φ}, {s, ϵ, R}], {r, ϵ, R}, PlotRange → {All, {0, 4}}]
{F1[1, 1.99], F1[1, 2], F1[1, 2.01]}
```



The k=1 component of the spectrum: Taylor expansion approach

```
v0[r_] := a + a2 r^2 + a3 r^3 + a4 r^4 + a5 r^5
H[r_] := v0''[r] + v0'[r]/r + e^(v0[r] - 1/2 r^2)
Solve[{Limit[H[r], r → 0] == 0, Limit[H'[r], r → 0] == 0,
    Limit[H''[r], r → 0] == 0, Limit[H'''[r], r → 0] == 0}, {a2, a3, a4, a5}][[1]];
v0[r] /. %
v0[r_] := a - (e^a r^2)/4 + 1/64 e^a (2 + e^a) r^4
```

$$a - \frac{e^a r^2}{4} + \frac{1}{64} e^a \left(2 + e^a\right) r^4$$

```
f0[r_] := r + a2 r^2 + a3 r^3 + a4 r^4 + a5 r^5
ψ0[r_] := α1 r + α2 r^2 + α3 r^3 + α4 r^4 + α5 r^5
H[r_] := f0''[r] + (r f0'[r] - f0[r])/r^2 + (v0'[r] - r) (f0'[r] - ψ0'[r]) + (λ + e^(v0[r] - 1/2 r^2)) f0[r]
HH[r_] := ψ0''[r] + (r ψ0'[r] - ψ0[r])/r^2 + e^(v0[r] - 1/2 r^2) f0[r]
{Limit[H[r], r → 0] == 0, Limit[HH[r], r → 0] == 0,
  Limit[H'[r], r → 0] == 0, Limit[HH'[r], r → 0] == 0, Limit[H''[r], r → 0] == 0,
  Limit[HH''[r], r → 0] == 0, Limit[H'''[r], r → 0] == 0, Limit[HH'''[r], r → 0] == 0};
Solve[%, {a2, a3, a4, a5, α1, α2, α3, α4}][[1]];
{r + a2 r^2 + a3 r^3 + a4 r^4 + a5 r^5, α1 r + α2 r^2 + α3 r^3 + α4 r^4 + α5 r^5} /. %
```

$$\Big\{ r + \frac{1}{4} e^{-a} r^3 \left(2 e^a + e^{2a} - 96\,\alpha 5\right) + r^5 \left(\frac{1}{384} e^a \left(14 + 5 e^a - 2\lambda\right) - \frac{1}{192} e^{-a} \left(2 e^a + e^{2a} - 96\,\alpha 5\right) \left(-6 + e^a + 2\lambda\right)\right),$$
$$-\frac{1}{8} e^a r^3 + r^5 \alpha 5 - \frac{e^{-a} r \left(6 e^a + 5 e^{2a} - 384\,\alpha 5 + 2 e^a \lambda\right)}{2 + e^a} \Big\}$$

The k=1 component of the spectrum: an ansatz

```
ϵ = 10^-6;
R = 7;

Mass[a_] :=
 2 π m[R] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]], m'[s] == s Exp[-1/2 s^2 + v[s]],
     v[ϵ] == a, v'[ϵ] == 0, m[ϵ] == 0}, {v, v', m}, {s, ϵ, R}][[1]]

F[a_, k_, λmin_, λmax_] :=
 Plot[Log[1 + f[R]^2] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
     -f''[s] - f'[s]/s + k^2 f[s]/s^2 == (v'[s] - s) (f'[s] - ψ'[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
     -ψ''[s] - ψ'[s]/s + k^2 ψ[s]/s^2 == f[s] Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0,
     f'[ϵ] == -(1/2 e^a + 1), f[ϵ] == -(1/2 e^a + 1) ϵ/k^2, ψ'[ϵ] == -1/2 e^a, ψ[ϵ] == -ϵ/(2 k^2) e^a},
    {v, v', f, f', ψ, ψ'}, {s, ϵ, R}], {λ, λmin, λmax}, PlotRange → {Automatic, All}]

F[1, 1, -1, 6]
```



```
η = 10^-8;
R = 7;
F[a_, λ_] :=
 Log[1 + f[R]^2] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]], -f''[s] - f'[s]/s + f[s]/s^2 ==
      (v'[s] - s) (f'[s] - ψ'[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s], -ψ''[s] - ψ'[s]/s + ψ[s]/s^2 ==
      f[s] Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0, f'[ϵ] == -(1/2 ⅇ^a + 1), f[ϵ] == -(1/2 ⅇ^a + 1) ϵ,
     ψ'[ϵ] == -1/2 ⅇ^a, ψ[ϵ] == -ϵ/2 ⅇ^a}, {v, v', f, f', ψ, ψ'}, {s, ϵ, R}][[1]]
Fiter[a_, λ_, h_, b_] := If[Abs[h] < η, {λ, h}, Module[{m = F[a, λ]},
   If[(m - b) < 0, Fiter[a, λ + h, h, m], Fiter[a, λ - h / 2, -h / 2, m]]]]
Search[a_, λ_, h_] := Fiter[a, λ, h, F[a, λ - h]]
Fpl[a_, λ_] :=
 Plot[Exp[-1/2 r^2 + v[r]] (f'[r]^2 + f[r]^2) /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
       -f''[s] - f'[s]/s + f[s]/s^2 == (v'[s] - s) (f'[s] - ψ'[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
       -ψ''[s] - ψ'[s]/s + ψ[s]/s^2 == f[s] Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0,
       f'[ϵ] == -(1/2 ⅇ^a + 1), f[ϵ] == -(1/2 ⅇ^a + 1) ϵ, ψ'[ϵ] == -1/2 ⅇ^a, ψ[ϵ] == -ϵ/2 ⅇ^a},
      {v, v', f, f', ψ, ψ'}, {s, ϵ, R}][[1]], {r, ϵ, R}]

Search[1, 0.8, 0.1]
Search[1, 3, 0.1]
λ2 = %[[1]];
R = 8;
Fpl[1, λ2]
```

$\{1., 5.96046 \times 10^{-9}\}$

$\{3.22762, 5.96046 \times 10^{-9}\}$



The k=1 component of the spectrum: general case, a first method

```
ϵ = 10^-6;
R = 3;
F[a_, λ_, pmin_, pmax_] :=
 Plot[Log[1 + m[R]^2] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
     -f''[s] - f'[s]/s + f[s]/s^2 == (v'[s] - s) (f'[s] - ψ'[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
     -ψ''[s] - ψ'[s]/s + ψ[s]/s^2 == f[s] Exp[-1/2 s^2 + v[s]],
     m'[s] == s Exp[-1/2 s^2 + v[s]] (f'[s]^2 + f[s]^2), v[ϵ] == a, v'[ϵ] == 0,
     f'[ϵ] == -1, f[ϵ] == -ϵ, ψ'[ϵ] == -p, ψ[ϵ] == -ϵ p, m[ϵ] == 0},
    {v, v', f, f', ψ, ψ', m}, {s, ϵ, R}], {p, pmin, pmax}, PlotRange → All]
{F[1, 0.5, 0, 3], F[1, 1, 0, 3], F[1, 3, -1, 3]}
```



```
F[1, 1, 0, 3]
```



```
η = 10^-6;
Fp[a_, λ_, p_] :=
 Log[1 + m[R]^2] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]], -f''[s] - f'[s]/s + f[s]/s^2 ==
      (v'[s] - s) (f'[s] - ψ'[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s], -ψ''[s] - ψ'[s]/s + ψ[s]/s^2 ==
      f[s] Exp[-1/2 s^2 + v[s]], m'[s] == s Exp[-1/2 s^2 + v[s]] (f'[s]^2 + f[s]^2), v[ϵ] == a,
     v'[ϵ] == 0, f'[ϵ] == -1, f[ϵ] == -ϵ, ψ'[ϵ] == -p, ψ[ϵ] == -ϵ p, m[ϵ] == 0},
    {v, v', f, f', ψ, ψ', m}, {s, ϵ, R}][[1]]
Fpiter[a_, λ_, p_, h_, b_] := If[Abs[h] < η, {p, h}, Module[{m = Fp[a, λ, p]},
   If[(m - b) < 0, Fpiter[a, λ, p + h, h, m], Fpiter[a, λ, p - h / 2, -h / 2, m]]]]
Searchp[a_, λ_, p_, h_] := Fpiter[a, λ, p, h, Fp[a, λ, p - h]]
popt[a_, λ_] := Searchp[a, λ, 0.1, 0.1][[1]]
```

```
Plot[popt[1, λ], {λ, -1, 6}]
```




```
F[a_, λmin_, λmax_] := Plot[
  Module[{p = popt[a, λ]}, Log[1 + f[R]^2] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
      -f''[s] - f'[s]/s + f[s]/s^2 == (v'[s] - s) (f'[s] - ψ'[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
      -ψ''[s] - ψ'[s]/s + ψ[s]/s^2 == f[s] Exp[-1/2 s^2 + v[s]],
      m'[s] == s Exp[-1/2 s^2 + v[s]] (f'[s]^2 + f[s]^2), v[ϵ] == a, v'[ϵ] == 0, f'[ϵ] == -1,
      f[ϵ] == -ϵ, ψ'[ϵ] == -p, ψ[ϵ] == -ϵ p, m[ϵ] == 0}, {v, v', f, f', ψ, ψ', m}, {s, ϵ, R}]],
  {λ, λmin, λmax}, PlotRange → {Automatic, All}]
```

```
F[0.1, -1, 6]
```



```
Fk[a_, λ_] :=
 Module[{p = popt[a, λ]}, Log[1 + f[R]^2] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
      -f''[s] - f'[s]/s + f[s]/s^2 == (v'[s] - s) (f'[s] - ψ'[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
      -ψ''[s] - ψ'[s]/s + ψ[s]/s^2 == f[s] Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0, f'[ϵ] == -1,
      f[ϵ] == -ϵ, ψ'[ϵ] == -p, ψ[ϵ] == -ϵ p}, {v, v', f, f', ψ, ψ'}, {s, ϵ, R}]][[1]]
Fkiter[a_, λ_, h_, b_] := If[Abs[h] < η, {λ, h}, Module[{m = Fk[a, λ]},
   If[(m - b) < 0, Fkiter[a, λ + h, h, m], Fkiter[a, λ - h / 2, -h / 2, m]]]]
Searchk[a_, λ_, h_] := Fkiter[a, λ, h, Fk[a, λ - h]]
Bifurcationk[λ_, h_] := ParametricPlot[{Mass[a], Searchk[a, λ, h][[1]]}, {a, -3, 10}]
```

```
F[1, 0.8, 0.9]
Searchk[1, 0.5, 0.1]
```



$\{0.839176, -7.62939 \times 10^{-7}\}$

```
R = 2.75;
T1 = {0, 0.1, 0.25, 0.5, 0.75, 1, 1.5, 2, 2.5, 3};
Table[{T1[[k]], Searchk[T1[[k]], 0.5, 0.1][[1]]}, {k, 1, Length[T1]}];
ListLinePlot[%, PlotRange → All]
```



The k=1 component of the spectrum: general case, an accurate method

```
η = 10^-8;
R = 7;
Fp[a_, λ_, p_] := (m[R] + 2 p)^2 /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
    -f''[s] - f'[s]/s + f[s]/s^2 == (v'[s] - s) (f'[s] - ψ'[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
    -ψ''[s] - ψ'[s]/s + ψ[s]/s^2 == f[s] Exp[-1/2 s^2 + v[s]], m'[s] == Exp[-1/2 s^2 + v[s]] f[s],
    v[ϵ] == a, v'[ϵ] == 0, f'[ϵ] == -1, f[ϵ] == -ϵ, ψ'[ϵ] == -p, ψ[ϵ] == -ϵ p, m[ϵ] == 0},
   {v, v', f, f', ψ, ψ', m}, {s, ϵ, R}][[1]]
Fpiter[a_, λ_, p_, h_, b_] := If[Abs[h] < η, {p, h}, Module[{m = Fp[a, λ, p]},
   If[(m - b) < 0, Fpiter[a, λ, p + h, h, m], Fpiter[a, λ, p - h / 2, -h / 2, m]]]]
Searchp[a_, λ_, p_, h_] := Fpiter[a, λ, p, h, Fp[a, λ, p - h]]
popt[a_, λ_] := Searchp[a, λ, 0.1, 0.1][[1]]
```

```
Plot[Fp[1, 1, p], {p, 0, 2}]
popt[1, 1]
```




```
0.576117
```

```
N[e/(e + 2)]
```

```
0.576117
```

```
Plot[popt[1, λ], {λ, -1, 6}]
```



```
F[a_, λmin_, λmax_] := Plot[
  Module[{p = popt[a, λ]}, Log[1 + f[R]^2] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
      -f''[s] - f'[s]/s + f[s]/s^2 == (v'[s] - s) (f'[s] - ψ'[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
      -ψ''[s] - ψ'[s]/s + ψ[s]/s^2 ==  f[s] Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0, f'[ϵ] == -1,
      f[ϵ] == -ϵ, ψ'[ϵ] == -p, ψ[ϵ] == -ϵ p}, {v, v', f, f', ψ, ψ'}, {s, ϵ, R}]],
  {λ, λmin, λmax}, PlotRange → {Automatic, All}]
F[0.1, -1, 6]
```




```
Fk[a_, λ_] :=
 Module[{p = popt[a, λ]}, Log[1 + f[R]^2] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
      -f''[s] - f'[s]/s + f[s]/s^2 == (v'[s] - s) (f'[s] - ψ'[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
      -ψ''[s] - ψ'[s]/s + ψ[s]/s^2 ==  f[s] Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0, f'[ϵ] == -1,
      f[ϵ] == -ϵ, ψ'[ϵ] == -p, ψ[ϵ] == -ϵ p}, {v, v', f, f', ψ, ψ'}, {s, ϵ, R}]][[1]]
Fkiter[a_, λ_, h_, b_] := If[Abs[h] < η, {λ, h}, Module[{m = Fk[a, λ]},
   If[(m - b) < 0, Fkiter[a, λ + h, h, m], Fkiter[a, λ - h / 2, -h / 2, m]]]]
Searchk[a_, λ_, h_] := Fkiter[a, λ, h, Fk[a, λ - h]]
Bifurcationk[λ_, h_] := ParametricPlot[{Mass[a], Searchk[a, λ, h][[1]]}, {a, -3, 5}]
{Searchk[0.1, 0.5, 0.1], Searchk[1, 0.5, 0.1], Searchk[5, 0.5, 0.1]}
```

$\{\{1., 5.96046\times10^{-9}\}, \{1., 5.96046\times10^{-9}\}, \{1., 5.96046\times10^{-9}\}\}$

```
{Searchk[0.1, 3, 0.1], Searchk[1, 3, 0.1], Searchk[5, 3, 0.1]}
```

$\{\{3.11374, 5.96046\times10^{-9}\}, \{3.22762, 5.96046\times10^{-9}\}, \{4.02916, 5.96046\times10^{-9}\}\}$

```
Off[InterpolatingFunction]
```

```
Pk1 = ListLinePlot[
   Table[{Mass[a], Searchk[a, 0.5, 0.1][[1]]}, {a, -3, 5, 0.25}], PlotStyle → Red];
Show[Pk1, AspectRatio → 1]
```




```
Pk2 = ListLinePlot[
   Table[{Mass[a], Searchk[a, 3, 0.2][[1]]}, {a, -3, 10, 0.25}], PlotStyle → Red];
Show[Pk2, AspectRatio → 1]
```



```
Pk3 = ListLinePlot[
   Table[{Mass[a], Searchk[a, 5.5, 0.2][[1]]}, {a, -3, 10, 0.25}], PlotStyle → Red];
Show[Pk3, AspectRatio → 1]
```




```
Show[{Pk1, Pk2, Pk3}, PlotRange → {{0, 8 π}, {0, 7.5}},
 AspectRatio → 1, AxesOrigin → {0, 0}]
```



```
Show[{P1, P2, P3, Pk1, Pk2, Pk3},
 PlotRange → {{0, 8 π}, {0, 7.5}}, AspectRatio → 1, AxesOrigin → {0, 0}]
```




```
Pkl1 = ListLinePlot[{{0, 1}, {8 π, 1}}, PlotStyle → Red];
Pl1 = ListLinePlot[{{0, 2}, {8 π, 2}}];
```

```
Show[{Pl1, P2, P3, Pkl1, Pk2, Pk3},
 PlotRange → {{0, 8 π}, {0, 7.5}}, AspectRatio → 1, AxesOrigin → {0, 0}]
```



The spectrum: general case, an accurate method

```
η = 10^-8;
R = 7;
Fp[a_, k_, λ_, p_] := (m[R] + 2 p)^2 /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
    -f''[s] - f'[s]/s + k^2 f[s]/s^2 == (v'[s] - s) (f'[s] - ψ'[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
    -ψ''[s] - ψ'[s]/s + k^2 ψ[s]/s^2 == f[s] Exp[-1/2 s^2 + v[s]], m'[s] == Exp[-1/2 s^2 + v[s]] f[s],
    v[ϵ] == a, v'[ϵ] == 0, f'[ϵ] == -1, f[ϵ] == -ϵ, ψ'[ϵ] == -p, ψ[ϵ] == -ϵ p, m[ϵ] == 0},
   {v, v', f, f', ψ, ψ', m}, {s, ϵ, R}][[1]]
Fpiter[a_, k_, λ_, p_, h_, b_] := If[Abs[h] < η, {p, h}, Module[{m = Fp[a, k, λ, p]},
   If[(m - b) < 0, Fpiter[a, k, λ, p + h, h, m], Fpiter[a, k, λ, p - h / 2, -h / 2, m]]]]
Searchp[a_, k_, λ_, p_, h_] := Fpiter[a, k, λ, p, h, Fp[a, k, λ, p - h]]
popt[a_, k_, λ_] := Searchp[a, k, λ, 0.1, 0.1][[1]]
```

```
Plot[Fp[1, 1, 1, p], {p, 0, 2}]
popt[1, 1, 1]
```



0.576117

```
Plot[Fp[1, 2, 1, p], {p, 0, 2}]
popt[1, 2, 1]
```



1.15701

```
Plot[popt[1, 1, λ], {λ, -1, 6}]
```




```
F[a_, k_, λmin_, λmax_] := Plot[Module[{p = popt[a, k, λ]},
    Log[1 + f[R]^2] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
        -f''[s] - f'[s]/s + k^2 f[s]/s^2 == (v'[s] - s) (f'[s] - ψ'[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
        -ψ''[s] - ψ'[s]/s + k^2 ψ[s]/s^2 == f[s] Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0,
        f'[ϵ] == -1, f[ϵ] == -ϵ, ψ'[ϵ] == -p, ψ[ϵ] == -ϵ p}, {v, v', f, f', ψ, ψ'}, {s, ϵ, R}]],
   {λ, λmin, λmax}, PlotRange → {Automatic, All}]
```

```
F[0.1, 1, -1, 6]
```



```
F[0.1, 2, -1, 6]
```



```
F[0.1, 2, 2, 10]
```



```
Fk[a_, k_, λ_] := Module[{p = popt[a, k, λ]},
   Log[1 + f[R]^2] /. NDSolve[{-v''[s] - v'[s]/s == Exp[-1/2 s^2 + v[s]],
      -f''[s] - f'[s]/s + k^2 f[s]/s^2 == (v'[s] - s) (f'[s] - ψ'[s]) + (Exp[-1/2 s^2 + v[s]] + λ) f[s],
      -ψ''[s] - ψ'[s]/s + k^2 ψ[s]/s^2 == f[s] Exp[-1/2 s^2 + v[s]], v[ϵ] == a, v'[ϵ] == 0, f'[ϵ] == -1,
      f[ϵ] == -ϵ, ψ'[ϵ] == -p, ψ[ϵ] == -ϵ p}, {v, v', f, f', ψ, ψ'}, {s, ϵ, R}]][[1]]
Fkiter[a_, k_, λ_, h_, b_] := If[Abs[h] < η, {λ, h}, Module[{m = Fk[a, k, λ]},
   If[(m - b) < 0, Fkiter[a, k, λ + h, h, m], Fkiter[a, k, λ - h / 2, -h / 2, m]]]]
Searchk[a_, k_, λ_, h_] := Fkiter[a, k, λ, h, Fk[a, k, λ - h]]
Bifurcationk[k_, λ_, h_] :=
 ParametricPlot[{Mass[a], Searchk[a, k, λ, h][[1]]}, {a, -3, 5}]
```

```
{Searchk[0.1, 1, 0.5, 0.1], Searchk[1, 1, 0.5, 0.1], Searchk[5, 1, 0.5, 0.1]}
```

$\{\{1., 5.96046\times10^{-9}\}, \{1., 5.96046\times10^{-9}\}, \{1., 5.96046\times10^{-9}\}\}$

```
{Searchk[0.1, 1, 3, 0.1], Searchk[1, 1, 3, 0.1], Searchk[5, 1, 3, 0.1]}
```

$\{\{3.11374, 5.96046\times10^{-9}\}, \{3.22762, 5.96046\times10^{-9}\}, \{4.02916, 5.96046\times10^{-9}\}\}$

```
Off[InterpolatingFunction]
```

```
Pkk1 = ListLinePlot[
   Table[{Mass[a], Searchk[a, 2, 4, 0.1][[1]]}, {a, -3, 10, 0.25}], PlotStyle → Brown];
Show[Pkk1, AspectRatio → 1]
```




```
Pkk2 = ListLinePlot[
   Table[{Mass[a], Searchk[a, 2, 6, 0.2][[1]]}, {a, -3, 10, 0.25}], PlotStyle → Brown];
Show[Pkk2, AspectRatio → 1]
```



```
Pkk3 = ListLinePlot[
   Table[{Mass[a], Searchk[a, 2, 8, 0.2][[1]]}, {a, -3, 10, 0.25}], PlotStyle → Brown];
Show[Pkk3, AspectRatio → 1]
```




```
Show[{Pkk1, Pkk2, Pkk3}, PlotRange → {{0, 8 π}, {0, 9}},
 AspectRatio → 1, AxesOrigin → {0, 0}]
```




```
Pkl1 = ListLinePlot[{{0, 1}, {8 π, 1}}, PlotStyle → Red];
Pl1 = ListLinePlot[{{0, 2}, {8 π, 2}}];
```

The spectrum

```
Show[{Pl1, P2, P3, Pkl1, Pk2, Pk3, Pkk1, Pkk2, Pkk3},
 PlotRange → {{0, 8 π}, {0, 9}}, AspectRatio → 1, AxesOrigin → {0, 0}]
```



```
Show[%, PlotRange → {{0, 8 π}, {0, 7}}, AspectRatio → 1, AxesOrigin → {0, 0}]
```